
144. Loginis kubas. Norėdami pasiekti tašką Z, turėsime eiti arba per viršūnę a, arba per viršūnę d, arba per f. Taigi loginis reiškinys turės būti šitoks:

(a and ?) or (d and ?) or (f and ?)

Jei esame viršūnėje a, tašką Z galime pasiekti arba eidami per viršūnę, b arba per c. Todėl reiškinio dalį (a and ?) reikėtų užrašyti šitaip: (a and (b or c)). Analogiškai įrašome reiškinius vietoj kitų likusių klaustukų.

Teisingas sprendimas yra reiškinys:

a and (b or c) or d and (b or e) or f and (c or e)

Gali būti ir kitoks šiam reiškiniui ekvivalentus sprendinys.



Figūros iš virvutės. Kvadratą sudaryti galima tik tada, kai mazgelių skaičius yra dalus iš keturių. Kadangi kraštinių ilgis vienodas, tai arba galime išlankstyti vieną kvadratą, arba iš viso negalime, taigi kvadratų skaičius lygus arba nuliui, arba vienam.

Stačiakampį galima sudaryti, kai mazgelių skaičius dalus iš dviejų (ir būtinai didesnis kaip du). Jei mazgelių yra n, tuomet dviejų stačiakampio kraštinių, turinčių bendrą viršūnę, ilgių suma lygi n div 2. Trumpesniosios stačiakampio kraštinės ilgis gali kisti nuo 1 iki (n div 2) div 2. Taigi, jei mazgelių skaičius dalus iš dviejų, galima sudaryti n div 4 stačiakampių. Iš šio skaičiaus reikėtų atimti sudarytų kvadratų skaičių. Mat skaičiuojant pagal šią formulę kvadratas laikomas stačiakampiu ir jis bus prie jų prisumuotas.

Norėdami sužinoti, kiek galima išlankstyti skirtingų trikampių, turėsime perrinkti visus įmanomus virvutės padalinimo į tris dalis variantus. 

Kai virvutės dalių ilgiai yra a, b ir c, tai iš jų galima sudaryti trikampį tik tuomet, jei teisingos šios nelygybės:

a + b > c, a + c > b, ir b + c > a

Sudarome procedūrą.

procedure figūros (n: integer; 

                   var trik, kvadr, stač: integer);

  var a, b, c: integer;

begin

  { rasime kvadratų skaičių }

  if n mod 4 = 0

     then kvadr := 1

     else kvadr := 0;

  { rasime stačiakampių skaičių }

  if (n mod 2 = 0) and (n > 2)

     then stač := n div 4 - kvadr

     else stač := 0;

  { rasime trikampių skaičių }

  trik := 0;

  for a := 1 to n do

    for b := a to n-a do

      begin

        c := n - (a + b);

        if (c >= a) and (c >= b) { vienodi trikampiai nesikartoja } and 

           (a + b > c) and (a + c > b) and (b + c > a)

           then trik := trik + 1

      end;

end; { figūros }





Testo nr.�Pradinis duomuo�Rezultatai�Pastabos��1�2�Trikampių skaičius: 0

Kvadratų skaičius: 0

Stačiakampių skaičius: 0�Mazgelių skaičius dalus Iš dviejų, tačiau mazgelių per mažai��2�3�Trikampių skaičius: 1

Kvadratų skaičius: 0

Stačiakampių skaičius: 0�Galima sudaryti vienintelį trikampį��3�4�Trikampių skaičius: 0

Kvadratų skaičius: 1

Stačiakampių skaičius: 0�Galima sudaryti vienintelį kvadratą��4�13�Trikampių skaičius: 5

Kvadratų skaičius: 0

Stačiakampių skaičius: 5�Tikrinama, ar teisingai randamas trikampių skaičius; rastieji trikampiai – (1, 6, 6), (2, 5, 6), (3, 4, 6), (3, 5, 5), (4, 4, 5)��5�14�Trikampių skaičius: 4

Kvadratų skaičius: 0

Stačiakampių skaičius: 3�Tikrinama, ar teisingai randamas stačiakampių skaičius��6�16�Trikampių skaičius: 5

Kvadratų skaičius: 1

Stačiakampių skaičius: 3�Tikrinama, ar kvadratas nepriskaičiuojamas ir prie stačiakampių��7�232�Trikampių skaičius: 1121

Kvadratų skaičius: 1

Stačiakampių skaičius: 57�Didesnis atsitiktinis testas��

146. Robotas ir kubeliai. Tarkime, kad kiekvienoje iš trijų krūvelių yra k1, k2, k3 kubelių. Akivaizdu, kad perstačius kubelius kiekvienoje krūvelėje turės būti arba k ( (k1 + k2 + k3) div 3, arba k + 1 kubelis (tuo atveju, jei kubelių skaičius nesidalija iš 3). 

� EMBED Visio.Drawing.5  ���

Likusių vieno arba dviejų kubelių negalima dėlioti bet kaip. Žemiau parodyti atvejai nėra sprendiniai. 

� EMBED Visio.Drawing.5  ���

Jei lieka vienas kubelis, jis būtinai turi būti ant vienos iš kraštinių krūvelių. Kubelis paliekamas (arba padedamas) ant tos kraštinės krūvelės, kurioje yra daugiau kubelių.

Jei lieka du kubeliai, vienas jų būtinai padedamas (arba lieka) ant viduriniosios krūvelės, antras – ant tos kraštinės krūvelės, kurioje yra daugiau kubelių.

Taigi pirmas algoritmo žingsnis – reikia apskaičiuoti, kiek po perkilnojimo kiekvienoje krūvelėje bus kubelių.

Kubelius perkilnojame taip. Imame pirmą krūvelę ir jei joje yra daugiau negu turėtų būti kubelių, atliekamus kubelius perkeliame ant antros krūvelės. Tą patį padarome su trečiąja krūvele. Na, o atliekamus kubelius iš antros krūvelės dedame į pirmą ar į trečią krūvelę, priklausomai nuo to, kurioje krūvelėje yra mažiau kubelių. 

Dėlioti visuomet reikia tokia tvarka (pirma nuo pirmos ir trečios krūvelės, o tik po to nuo antros). Mat pradėję dėlioti nuo antros krūvelės galime pamiršti patikrinti, ar nereikia antrą kartą kelti kubelių nuo antros krūvelės. Pasižiūrėkime į žemiau esančius paveikslėlius. Perkilnojus kubelius kiekvienoje krūvelėje turės būti po tris kubelius. Tačiau, jei pirma vieną kubelį perkelsime nuo antros ant trečios krūvelės, o paskui – nuo pirmos ant antros, tai teks dar kartą kelti nuo antros ant trečios krūvelės. 

� EMBED Visio.Drawing.5  ���

Dėliojant pagal aukščiau aprašytą algoritmą žingsnių skaičius visada bus mažiausias, nes nebus atliekamas nei vienas nereikalingas perkėlimas.



procedure kilnoti_kubelius (k1, k2, k3: integer);



  procedure kelti (x, y, kub: integer);

  { perkelia kub kubelių nuo krūvelės x ant krūvelės y }

    var i: integer;

  begin

    for i := 1 to kub do

      writeln (x, ' ', y);

  end; { kelti }



  var kub, liko, turi1, turi2, turi3: integer;

begin

  { rasime kubelių skaičių kiekvienoje krūvelėje }

  kub := (k1 + k2 + k3) div 3;

  liko := (k1 + k2 + k3) mod 3;

  turi1 := kub; turi2 := kub; turi3 := kub;

  if liko > 0

     then if k1 < k3

             then turi3 := turi3 + 1

             else turi1 := turi1 + 1;

  if liko = 2

     then turi2 := turi2 + 1;

  { perkilnojame kubelius, jei juos reikia perkelti }

  if (k1 = turi1) and (k2 = turi2) and (k3 = turi3)

     then writeln ('GERAI')

     else begin

            if k1 > turi1 then begin

                                 kelti (1, 2, k1 - turi1);

                                 k2 := k2 + (k1-turi1);

                               end;

            if k3 > turi3 then begin

                                 kelti (3, 2, k3 - turi3);

                                 k2 := k2 + (k3-turi3)

                               end;

            if k2 > turi2

                    then begin

                           if k1 < turi1

                              then kelti (2, 1, turi1 - k1);

                           if k3 < turi3

                              then kelti (2, 3, turi3 - k3);

                         end;

          end;

end; { kilnoti_kubelius }



Testo nr.�Pradiniai duomenys�Vienas galimų sprendinių�Mažiausias žingsnių skaičius�

Pastabos��1�5 2 5�1 2

3 2�2�Kubelių skaičius dalus iš trijų, vidurinėje krūvelėje mažiausia kubelių��2�2 7 3�2 1

2 1

2 3�3�Kubelių skaičius dalus iš trijų, vidurinėje krūvelėje daugiausia kubelių��3�9 5 1�1 2

1 2

1 2

1 2

2 3

2 3

2 3

2 3�8�Kubelių skaičius dalus iš trijų, pirmoje krūvelėje daugiau nei antrojoje, antrojoje – daugiau nei trečiojoje��4�7 3 6�1 2

3 2�2�Kubelių skaičius nesidalija iš 3, likutis – 1, vidurinėje krūvelėje mažiausia kubelių��5�3 7 3�2 1

2 1

2 3�3�Kubelių skaičius nesidalija iš 3, likutis – 1, vidurinėje krūvelėje daugiausia kubelių��6�2 3 5�3 2

2 1�2�Kubelių skaičius nesidalija iš 3, likutis – 1 trečiojoje krūvelėje daugiau nei antrojoje, antrojoje – daugiau nei pirmojoje��7�5 6 5�2 1�1�Kubelį nuo vidurinės krūvelės reikia perkelti ant kraštinės��8�5 1 2�1 2

1 2�2�Kubelių skaičius nesidalija iš 3, likutis – 2, vidurinėje krūvelėje mažiausia kubelių��9�1 8 2�2 1

2 1

2 3

2 3�4�Kubelių skaičius nesidalija iš 3, likutis – 2, vidurinėje krūvelėje daugiausia laiptelių��10�2 2 7�3 2

3 2

3 2

2 1�4�Kubelių skaičius nesidalija iš 3, likutis – 2 trečiojoje krūvelėje daugiau nei antrojoje, antrojoje – tiek pat kiek pirmojoje��11�6 5 6�3 2�1�Kubelį nuo kraštinės krūvelės reikia perkelti ant vidurinės��12�3 3 4�GERAI�0�Nereikia perkelti nei vieno kubelio��

Pastaba. Sprendžiant šį uždavinį kubelių dėliojimo tvarka nebūtinai turi sutapti su čia pateikta, tačiau žingsnių skaičius turi būti toks pat.





Vyresniųjų grupė

147. Loginis kubas. Sudarome dvimatę lentelę k, kuri rodytų iš kurių į kurias viršūnes galime eiti. Pirmiausia lentelę užpildome pagal sąlygoje duotą brėžinį (t. y. sujungiame viršūnę a su viršūnėmis b, d, e, viršūnę b – su viršūnėmis a, c, f ir t. t.).

Po to lentelę papildome remdamiesi pradiniais duomenimis: t. y. jei viršūne eiti draudžiama, tai į ją negalima ateiti ir iš jos negalima išeiti. Pavyzdžiui, jei draudžiama eiti viršūne a, tai negalime pakliūti nei iš a į b, nei iš b į a. 

Panagrinėję kubą pamatysime, kad ieškomas kelias gali eiti ne daugiau kaip per tris tarpines viršūnes. Pavyzdžiui, kelias iš a į c gali eiti per viršūnes e, f, g. Vadinasi, jei iš pradinės viršūnės galima pasiekti galinę, tai tam prireiks daugiausia keturių žingsnių. 

Toliau remsimės tokia savybe: 

jei iš viršūnės x pasiekiama viršūnė y, o iš y – viršūnė z, tai iš x pasiekiama z, 

ir rasime visas viršūnes, pasiekiamas dviem, trim ir keturiais žingsniais.



type kubas = array ['a'..'h', 'a'..'h'] of boolean;

     viršūnės = array ['a'..'h'] of boolean;



function galima (prad, gal: char; v: viršūnės): boolean;

  var i, j: char;

      k: kubas;

      žingsn: integer;

begin

  { pradžioje laikome, kad nei per vieną viršūnę negalima eiti }

  for i := 'a' to 'h' do

    for j := 'a' to 'h' do

      k[i, j] := false;

  { sužymime iš kurių viršūnių į kurias galima pakliūti }

  k['a', 'b'] := true; k['b', 'a'] := true;

  k['a', 'd'] := true; k['d', 'a'] := true;

  k['a', 'e'] := true; k['e', 'a'] := true;

  k['c', 'b'] := true; k['b', 'c'] := true;

  k['c', 'd'] := true; k['d', 'c'] := true;

  k['c', 'g'] := true; k['g', 'c'] := true;

  k['f', 'b'] := true; k['b', 'f'] := true;

  k['f', 'e'] := true; k['e', 'f'] := true;

  k['f', 'g'] := true; k['g', 'f'] := true;

  k['h', 'e'] := true; k['e', 'h'] := true;

  k['h', 'd'] := true; k['d', 'h'] := true;

  k['h', 'g'] := true; k['g', 'h'] := true;

  { jei viršūne draudžiama eiti, į ją nei ateisi, nei iš jos išeisi }

  for i := 'a' to 'h' do

    if not v[i]

       then for j := 'a' to 'h' do

               begin

                 k[i, j] := false; k[j, i] := false;

               end;

  { surasime visas viršūnes, pasiekiamas iš pradinės viršūnės }

  { jei viršūnė pasiekiama, jai prireiks ne daugiau keturių žingsnių }

  for žingsn := 1 to 3 do

    for i := 'a' to 'h' do

      for j := 'a' to 'h' do

        if k[prad, i] and k[i, j]

           then begin

                  k[prad, j] := true;

                  k[j, prad] := true;

                end;

  galima := k[prad, gal]

end; { galima }



Testo nr.�Pradiniai duomenys 

(surašyti trijose eilutėse)�Rezultatas�Pastabos��1�d h 

false false false true 

false false false true�Galima�Pradinė ir galinė viršūnė gretimos, todėl galinė viršūnė visada pasiekiama��2�e g 

false false false false

true false true true�Galima�Yra vienintelis kelias (per viršūnę h), jungiantis pradinę ir galinę viršūnes ��3�e g 

true true true true

true false true false�Galima�Kelias eina per tris tarpines viršūnes��4�a g 

true false true false

true false true false�Negalima�Testas, kai kelio nėra��5�a g

true false true false

true false true true�Galima�Tas pats testas, kaip ir 4, tik leidžiama eiti per viršūnę h.��

148. Mokytojo diena. Norėdami sužinoti, kokia savaitės ar kokia mėnesio diena yra duotoji diena, pirmiausia reikia sužinoti, kokia savaitės diena yra duotųjų metų sausio 1-oji.

Žinoma, kad per metus sausio 1-oji pasislenka arba viena (jei metai paprastieji), arba dviem (jei metai keliamieji) dienomis. Pavyzdžiui, 1979 m. sausio 1-oji buvo pirmadienis. Kadangi 1979 metai nėra keliamieji, tai 1980 m. sausio 1-oji bus antradienį. Šie metai jau keliamieji, todėl 1981 m. sausio 1-oji pasislenka per dvi dienas – yra ketvirtadienį.

Kadangi ieškomi metai yra iš šio šimtmečio, tai atskaitos pradžią patogu pasirinkti šimtmečio pirmuosius arba paskutiniuosius metus. Žinome, kad 2000-ųjų metų sausio 1-oji yra šeštadienį. Eidami atgal ligi duotųjų metų, sužinome, kuria savaitės diena prasideda duotieji metai. 

Pastebėsime, kad jei metai nėra keliamieji, sausio mėnesio kalendorius sutampa su spalio mėnesio kalendoriumi, t. y. jei sausio 1-oji yra antradienis, tai ir spalio 1-oji bus taip pat antradienis. Jei metai keliamieji, spalio mėnesio kalendorius pasislenka per dieną į priekį: jei sausio 1-oji antradienis, tai spalio 1-oji – trečiadienis. Remdamiesi šiais samprotavimais, nesunkiai nustatome, kurią mėnesio dieną švenčiama Mokytojo diena.



function mok_diena (mt: integer): integer;

{ randą kurią mėnesio dieną duotaisiais metais švenčiama }

{ mokytojo diena }

  var i, diena, metai, posl: integer;



  function kel (m: integer): boolean;

  { ar metai m keliamieji }

  begin

    kel := (m mod 400 = 0) or

           (m mod 100 <> 0) and (m mod 4 = 0)

  end; { kel }



begin

  { nustatysime, kurią savaitės dieną prasideda duotieji metai }

  diena := 6; metai := 2000;

  for i := metai-1 downto mt do

    begin

      if kel (i) { randame poslinkį }

         then posl := 2

         else posl := 1;

      if diena - posl < 1

         then diena := diena + 7 - posl

         else diena := diena - posl;

    end;

  { rasime, kurią savaitės dieną prasideda spalis }

  if kel (mt)

     then if diena + 1 > 7

             then diena := 1

             else diena := diena + 1;

  { rasime kuri mėnesio diena yra pirmasis spalio sekmadienis }

  mok_diena := 7 - diena + 1

end; { mok_diena }



Testo nr.�Pradinis duomuo�Rezultatas�Kodėl pasirinktas toks testas��1�2000�spalio 1�Pradinis duomuo – paskutinieji šimtmečio metai��2�1998�spalio 4�Pradinis duomuo – šie metai (nesunku patikrinti atsakymo teisingumą)��3�1967�spalio 1�Mokytojo diena švenčiama pirmąją mėnesio dieną��4�1968�spalio 6�Duotieji metai yra keliamieji��5�1969�spalio 5�Duotieji metai yra paprastieji��

149. Laiptai iš kubelių. Šis uždavinys yra tipiškas rekursijos panaudojimo pavyzdys. Laiptų pakopas konstruojame iš kairės į dešinę. Sukonstruotas pakopas (t. y. jų aukščius) įrašome į vienmatę lentelę p. Taip pat įsimename ir žemiausios sukonstruotos pakopos numerį nr. 

Rekursinė funkcija (programoje pateikiame rekursinę procedūrą, kadangi vartojame išorinius kintamuosius) konstruoja tolesnę pakopą. 

Jei nebeliko nei vieno kubelio, tai reiškia, kad laiptai sukonstruoti. Išspausdiname sukonstruotų laiptų pakopų aukščius ir grįžtame vienu žingsniu atgal. 

Jei paskutiniosios pakopos aukštis lygus 1, o dar liko kubelių, vadinasi, priėjome aklavietę ir reikia sustabdyti konstravimą bei grįžti vienu žingsniu atgal. 

Priešingu atveju konstruojame tolesnę pakopą. Jos aukštis gali kisti nuo p[nr] iki 1, jei, žinoma, užtenka kubelių. Išbandę visus šiuos variantus, grįžtame žingsniu atgal.



program laiptai_iš_kubelių;



const MAX_N = 25; { maksimalus pakopų skaičius }

type pakopos = array [1..MAX_N] of integer;



{ žemiau surašyti globalieji kintamieji }

  var p: pakopos;

      nr: integer; { žemiausios pakopos numeris }





  procedure dėti (buvęs, { buvusio laiptelio aukštis }

                  liko: integer; { likusių kubelių skaičius }

                  var kiek: integer { skirtingų laiptų skaičius });

  { rekursinė procedūra, sudedanti vieną laiptelį }

    var i: integer;

  begin

    if liko = 0

       then begin

              kiek := kiek + 1; { sudėjome dar vieną variantą }

              for i := 1 to nr do

                write (p[i], ' ');

              writeln;

            end

       else if not ((buvęs = 1) and (liko > 0))

               then begin  { jei galima sudėti dar vieną laiptelį }

                      for i := buvęs-1 downto 1 do

                        if liko-i >= 0

                           then begin

                                  nr := nr + 1;

                                  p[nr] := i;

                                  dėti(i, liko-i, kiek);

                                  nr := nr - 1

                                end;

                    end;

  end; { dėti }



  var kiek, n, i: integer;

begin

  write ('Įveskite kubelių skaičių: ');

  readln (n);

  kiek := 0;

  nr := 1;

  for i := n downto 1 do

    begin

      p[nr] := i;

      dėti (i, n-i, kiek);

    end;

  writeln ('Laiptų skaičius: ', kiek);

end.



Testo nr.�Pradinis duomuo�Rezultatai��1�2�2

Laiptų skaičius: 1��

2�3�3

2 1

Laiptų skaičius: 2��





3�8�8

7 1

6 2

5 3

5 2 1

4 3 1

Laiptų skaičius: 6��





















4�14�14

13 1

12 2

11 3

11 2 1

10 4

10 3 1

9 5

9 4 1

9 3 2

8 6

8 5 1

8 4 2

8 3 2 1

7 6 1

7 5 2

7 4 3

7 4 2 1

6 5 3

6 5 2 1

6 4 3 1

5 4 3 2

Laiptų skaičius:22��

150. Elektroninis indikatorius. Uždavinys reikalauja daugiau kruopštumo negu loginio mąstymo. Kokia nors tvarka sunumeruojame brūkšnelius. Patogiau numeruoti ne viso indikatoriaus, o tik vieno skaitmens brūkšnelius. Tuomet išanalizuojame, kurie brūkšneliai turi degti, kai nuspaudžiamas atitinkamas skaitmuo, ir sukuriama lentelė, kurioje įsimenama ši informacija. Galiausiai peržiūrimi visi butų numeriai ir atitinkamai keičiamos brūkšnelių būsenos.

� EMBED Word.Picture.6  ���

Sudarysime lentelę, kurioje išnagrinėsime, kaip keičiasi brūkšnelio būsena visais įmanomais variantais.



Ankstesnė brūkšnelio būsena�Surinkus naują numerį brūkšnelis turi�Deja, indikatorius sugedęs, todėl brūkšnelis��degė�degti�nedega��degė�nedegti�dega��nedegė�degti�dega��nedegė�nedegti�nedega��

Matome, kad būsenos pakitimą nusako loginė operacija xor.



const BR = 7; { brūkšnelių skaičius vieno skaitmens indikatoriuje }

      MAX_N = 100; { maksimalus svečių skaičius }

type skaitmuo = array [1..BR] of boolean;

     indikatorius = array [1..2] of skaitmuo;

     skaitmenys = array [0..9] of skaitmuo;

     svečiai = array [1..MAX_N] of 1..99;



procedure ini_skaitmenys (var sk: skaitmenys);

{ inicializuoja masyvą skaitmenys }

  var aibė: array [0..9] of set of 1..BR;

      s, b: integer;

begin

  aibė[1] := [3, 4];           aibė[2] := [2, 3, 5, 6, 7];

  aibė[3] := [2, 3, 4, 5, 7];  aibė[4] := [1, 3, 4, 7];

  aibė[5] := [1, 2, 4, 5, 7];  aibė[6] := [1, 4, 5, 6, 7];

  aibė[7] := [2, 3, 4];   

  aibė[8] := [1, 2, 3, 4, 5, 6, 7];

  aibė[9] := [1, 2, 3, 4, 5, 7]; 

  aibė[0] := [1, 2, 3, 4, 5, 6];

  for s := 0 to 9 do

    for b := 1 to BR do

      sk[s, b] := b in aibė[s]

end; { ini_skaitmenys }



procedure sugedo (n: integer; sv: svečiai; 

                  var ind: indikatorius);

{ randa indikatoriaus būseną po svečių apsilankymo }

  var sk: skaitmenys;

      i, b, pirm, antr: integer;

begin

  { inicializuojame masyvą skaitmenys }

  ini_skaitmenys (sk);

  { pradžioje indikatorius rodė 00 }

  ind[1] := sk[0]; ind[2] := sk[0];

  { eina svečiai }

  for i := 1 to n do

    begin

      pirm := sv[i] div 10;

      antr := sv[i] mod 10;

      for b := 1 to BR do

        begin

          ind[1, b] := ind[1, b] xor sk[pirm, b];

          ind[2, b] := ind[2, b] xor sk[antr, b];

        end;

    end;

end; { sugedo }



Pateikiant pavyzdinį sprendimą, buvo patogiau butus, kuriuose lankėsi svečiai, saugoti masyve. Tačiau sąlygoje nėra svečių skaičiaus ribojimo. Todėl norint išvengti problemų, butų numerius galima įvedinėti tiesiogiai procedūroje sugedo ir pasinaudojus numeriu jo nebesaugoti.

Kadangi rezultatų spausdinimas nėra akivaizdus, pridedame procedūrą spausdinti. Ji turėtų būti kviečiama iš pagrindinės programos dalies.



procedure spausdinti (ind: indikatorius);

{ išveda į ekraną galutinę indikatoriaus būseną }

  var i, b, nr: integer;

      eil: array [1..3, 1..BR] of char;

begin

  { indikatorių spausdinsime trimis eilutėmis }

  for i := 1 to 7 do

    if i = 4

       then eil[1, i] := ' '

       else eil[1, i] := '.';

  eil[2] := eil[1]; eil[3] := eil[1];

  { suformuosime pirmą eilutę }

  if ind[1, 2] then eil[1, 2] := '_';

  if ind[2, 2] then eil[1, 6] := '_';

  { suformuosime antrąją eilutę }

  for i := 1 to 2 do

    for b := 1 to 3 do

      begin

        if b = 2

           then nr := 7

           else nr := b;

         if ind [i, nr]

            then if nr <> 7 then eil[2, (i-1)*4 + b] := 'I'

                            else eil[2, (i-1)*4 + b] := '_'

      end;

  { suformuosime trečiąją eilutę }

  for i := 1 to 2 do

   for b := 6 downto 4 do

    if ind[i, b]

       then if b <> 5  then eil[3, (i-1)*4 + 7-b] := 'I'

                       else eil[3, (i-1)*4 + 7-b] := '_';

  { spausdiname }

  writeln ('Indikatoriaus būsena');

   for i := 1 to 3 do

     writeln (eil[i]);

end; { spausdinti }

Pateiktų testų atsakymai



� EMBED Excel.Sheet.5  ���

151. Didmeninis pirkimas. Kojines galima pirkti trimis būdais:

1) pirkti tiksliai tiek, kiek reikia kojinių porų, stengiantis pirkti kuo daugiau dėžių ir pundelių;

2) nepirkti nei vienos poros kojinių (t. y. pirkti tik dėžes ir pundelius), stengiantis nupirkti kuo daugiau dėžių;

3) pirkti tik dėžes (nepirkti nei vieno pundelio ir nei vienos poros).

Apskaičiuojame, kiek kainuos perkant kiekvienu būdu ir išrenkame pigiausią. Beje, jei dėžę sudarytų kojinių poros (o ne pundeliai), tai šis sprendimas netiktų.



procedure pirkti (n, { kojinių porų skaičius }

                  dėžė, pora, pund: integer; { kainos }

                  var dė_sk, pu_sk, po_sk: integer 

                       { perkami kiekiai });

  var kaina1, kaina2, kaina3: integer;

begin

  { perkame tiksliai tiek, kiek reikia }

  dė_sk := n div (12 * 12);

  pu_sk := n mod (12 * 12) div 12;

  po_sk := n mod 12;

  { apskaičiuojame, kiek kainuos mūsų pirkinys }

  kaina1 := dė_sk * dėžė + pu_sk * pund + po_sk * pora;

  { apskaičiuojame pirkimo antru būdu kainą }

  if po_sk > 0

     then kaina2 := dė_sk * dėžė + (pu_sk + 1) * pund

     else kaina2 := maxint; { fiktyvi kaina }

  { apskaičiuojame pirkimo trečiu būdu kainą }

  if pu_sk + po_sk > 0

     then kaina3 := (dė_sk + 1) * dėžė

     else kaina3 := maxint;

  { rasime geriausią pirkimo variantą }

  if (kaina2 ( kaina1) and (kaina2 < kaina3)

     then begin

            po_sk := 0;

            pu_sk := pu_sk + 1;

          end;

  else if (kaina3 ( kaina2) and (kaina3 ( kaina1)

       then begin

              po_sk := 0; pu_sk := 0;

              dė_sk := dė_sk + 1;

            end;

end; { pirkti }

152. Slapta kalba. Šiuo uždaviniu patikrinama, ar sprendžiantysis sugeba dirbti su simbolių eilutėmis. Uždavinys nereikalauja ypatingesnio algoritmo. Imama po vieną abiejų kalbų žodį. Antrasis žodis padalijamas į dvi dalis, tos dalys sukeičiamos vietomis ir sujungiamos. Tikrinama, ar tokiu būdu negavome pirmojo žodžio. Taip išbandomi visi antrojo žodžio dalijimo į dvi dalis variantai. Radus pirmą klaidingai išverstą žodį darbas nutraukiamas.

Uždavinys turi keletą „kabliukų“. Reikia nepamiršti, kad žodis slaptoje kalboje gali sutapti su lietuvišku žodžiu tik tada, kai žodį sudaro viena raidė, pavyzdžiui, lietuviškas žodis kalnas verčiant negali būti paliktas tas pats.

Taip pat reikėtų nepamiršti, kad žodžiai gali būti sudaryti ir iš didžiųjų, ir iš mažųjų raidžių. Vertime didžiosios ir mažosios raidės gali pasikeisti. Todėl prieš lyginant du žodžius, būtina juos abu perrašyti tik didžiosiomis (arba tik mažosiomis raidėmis).

Na, o perrašant žodį didžiosiomis raidėmis reikia nepamiršti, kad funkcija Upcase netinka lietuviškoms raidėms.



procedure vertimas (liet, slap: string;

                    var teis: boolean; { ar teisingai išversta }

                    var žodis: string { klaidingas žodis });



  function didž (r: char): char;

  { raidę r perrašo didžiąją raide }

  begin

   if r in ['Ą', 'Č', 'Ę', 'Ė', 'Į', 'Š', 'Ų', 'Ū', 'Ž']

    then didž := r

    else if not (r in ['ą', 'č', 'ę', 'ė', 'į', 'š', 'ų', 'ū'])

            then didž := upcase (r)

            else case r of

                  'ą': didž := 'Ą';

                  'č': didž := 'Č';

                  'ę': didž := 'Ę';

                  'ė': didž := 'Ė';

                  'į': didž := 'Į';

                  'š': didž := 'Š';

                  'ų': didž := 'Ų';

                  'ū': didž := 'Ū';

                  'ž': didž := 'Ž';

                end

  end; { didž }



  function teis_žodis (lt, sl: string): boolean;

  { tikrina, ar teisingai išverstas žodis }

    var pradžia, pabaiga: string;

        i: integer;

        pavyko: boolean;

  begin

    { žodžius perrašome didžiosiomis raidėmis }

    for i := 1 to length(lt) do

      begin

         lt[i] := didž (lt[i]);

         sl[i] := didž (sl[i]);

      end;

    { žodžių, turinčių vieną raidę keisti nereikia }

    pavyko := (length(lt) = 1) and (lt = sl);

    i := 1;

    while not pavyko and (i < length(lt)) do

      begin

        { žodį suskaidome į dvi dalis }

        pradžia := copy (sl, 1, i);

        pabaiga := copy (sl, i+1, length(sl)-i);

        pavyko := lt = (pabaiga+pradžia);

        i := i + 1;

      end;

    teis_žodis := pavyko;

  end; { teis_žodis }



  var tarpas: integer;

      lt, sl: string;

begin

  { taškus sakinių gale pakeičiame tarpais }

  liet := copy (liet, 1, length(liet) - 1) + ' ';

  slap := copy (slap, 1, length(slap) - 1) + ' ';

  teis := true;

  while (length(liet) > 0) and teis do

    { kol neperžiūrėti visi žodžiai ir nerasta klaidų }

    begin

      { imame vieną lietuvišką žodį }

      tarpas :=  pos(' ', liet);

      lt := copy (liet, 1, tarpas-1);

      delete (liet, 1, tarpas);

      { imame vieną išverstą žodį }

      tarpas :=  pos(' ', slap);

      sl := copy (slap, 1, tarpas-1);

      delete (slap, 1, tarpas);

      if length(lt) <> length(sl)

         then teis := false

         else teis := teis_žodis (lt, sl)

    end;

  if not teis

     then žodis := lt;

end; { vertimas }

153. Paveiksliukas iš žvaigždučių. Kvadrato langeliai bus numeruojami nuo 1 iki n2. Visus skaičius iš šio intervalo reikia suskirstyti į pirminius ir sudėtinius. Šį kartą pasirinksime mažiau įprastą algoritmą. 

Pradžioje laikykime kad visi skaičiai yra pirminiai. Tuomet paeiliui pašaliname sudėtinius skaičius. Pavyzdžiui, n = 3. Galimų pirminių skaičių aibė kol kas lygi { 1, 2, 3, 4, 5, 6, 7, 8, 9 }. Vienetas nėra pirminis, jį pašaliname. Tuomet pašaliname dvejeto sandaugas: 2x2, 2x3, 2x4. Aibė sumažėja iki { 2, 3, 5, 7, 9 }. Tuomet išmetame visas trejeto sandaugas: 3x3. Gauname { 2, 3, 5, 7}. Sandauga 3x2 buvo pašalinta ankstesniu žingsniu. O didesnių nei 3 skaičių sandaugos jau viršys 32. 

Sunkesnioji uždavinio dalis: sunumeruoti kvadratėlius nurodyta tvarka. Ypatingo algoritmo čia nėra. Numeruojame apeidami langelius nurodyta tvarka. Svarbu teisingai sekti kada keičiasi ėjimo kryptis ir kada pasiekiami kraštiniai kvadratėliai. 



const MAX_N = 20; { maksimalus paveiksliuko dydis }

type piešinys = array [1..MAX_N, 1..MAX_N] of boolean;

     lentelė = array [1..MAX_N*MAX_N] of boolean;



procedure piešti (n: integer; var pav: piešinys);

  var pirm: lentele;

      i, j, nn: integer;

  

  procedure dėlioti_žvaigždutes (n: integer; pirm: lentelė;

                                 var pav: piešinys);

  { sudėlioja žvaigždutes vietoj pirminių skaičių }

    var x, k, i, j: integer;

  begin

    k := -1;   { k rodo kryptį (ar į kairę, ar į dešinę)}

    i := 1; j := 1; { i – eilutė, j – stulpelis }

    for x := 1 to n*n do

      begin

         pav[i, j] := pirm[x];

         if (k = -1) and ((i = n) or (j = 1))

            then  { jei priėjome kairįjį arba apatinį kraštą }

               begin

                 if i = n

                    then j := j + 1

                    else i := i + 1;

                 k := -k

               end

         else if (k = 1) and ((i = 1) or (j = n))

            then  { jei priėjome dešinįjį arba viršutinįjį kraštą }

              begin

                if j = n

                   then i := i + 1

                   else j := j + 1;

                k := -k

              end

         else begin

                i := i - k;

                j := j + k

              end

      end;

  end; { dėlioti_žvaigždutes }



begin { piešti }

  { randame pirminius skaičius }

  nn := n * n;

  for i := 1 to nn do

    pirm[i] := true;

  {išmetame sudėtinius skaičius}

  pirm[1] := false;

  for i := 2 to nn div 2 do

    for j := 2 to nn div i do

        pirm[i*j] := false;

   dėlioti_žvaigždutes (n, pirm, pav);

end; { piešti }

154. Trys sodininkai. Matematiškai sąlygą performuluoti galima taip: reikia rasti trijų stačiakampių susikirtimo plotą. Tai yra, plotą didžiausio stačiakampio, priklausančio visiems trims duotiesiems stačiakampiams.

Pradėkime ne nuo stačiakampių, o nuo atkarpų. Duotos trys atkarpos. Reikia rasti jų sankirtą. 

154.2 pav.

Trijų atkarpų sankirta gali būti atkarpa, taškas arba visos trys atkarpos gali neturėti bendrų taškų. 

154.3 pav.

Matome, kad paties kairiausio taškas priklausančio visoms trims atkarpoms koordinatė lygi x1 = max (a1, b1, c1). Tuo tarpu paties dešiniausio taško koordinatė lygi x2 = min (a2, b2, c2). 

Jei x1 < x2, trijų atkarpų sankirta yra atkarpa, jei x1 = x2 – taškas ir galiausiai jei x1 > x2 – nėra taškų bendrų visoms trims atkarpoms.

Norint rasti trijų stačiakampių sankirtą, reikia rasti tų stačiakampių pagrindų bei šoninių kraštinių sankirtas. Gautosios atkarpos ir bus ieškomo stačiakampio kraštinės. Na o stačiakampio plotas lygus jo kraštinių ilgių sandaugai. 

154.4 pav.

Jei pagrindų ar šonininių kraštinių sankirta bus lygi taškui ar jos visai nesikirs, tai reikš, kad neegzistuoja stačiakampis priklausantis visiems trims duotiesiems stačiakampiams.



const DUOM = 'SODAS.DAT';

type  taškas = record

                 x, y: integer

               end;

      sklypas = record { apatinio ir viršutinio kampo koordinatės }

                  ap, vi: taškas

                end;



function min (a, b, c: integer): integer;

{ randa mažiausią iš trijų skaičių }

begin

  if (a <= b) and (a <= c)

     then min := a

     else if b <= c then min := b

                    else min := c

end; { min }



function max (a, b, c: integer): integer;

{ randa didžiausią iš trijų skaičių }

begin

  if (a >= b) and (a >= c)

     then max := a

     else if b >= c then max := b

                    else max := c

end; { max }



function vaismedžiai (s1, s2, s3: sklypas): integer;

{ kelis vaismedžius prižiūrės visi sodininkai }

  var b: sklypas;

begin

  { randame sklypo, priklausančio visiems trims sklypams, koordinates }

  b.ap.x := max (s1.ap.x, s2.ap.x, s3.ap.x);

  b.ap.y := max (s1.ap.y, s2.ap.y, s3.ap.y);

  b.vi.x := min (s1.vi.x, s2.vi.x, s3.vi.x);

  b.vi.y := min (s1.vi.y, s2.vi.y, s3.vi.y);

  if (b.vi.x < b.ap.x) or (b.vi.y < b.vi.y)

      then vaismedžiai := 0

      else vaismedžiai := (b.vi.x - b.ap.x) * (b.vi.y - b.ap.y)

end; { vaismedžiai }

Testus iliustruoja žemiau pateikti paveikslėliai.

Teksto tvarkymas. Sprendimą skaidysime į dvi tarpusavyje nesusijusias dalis: teksto sutvarkymą ir jau sutvarkyto teksto suskaidymą į eilutes po n simbolių. 

Tekstą sutvarkyti galima taip:

– skaitant pradinius duomenis tekstas (jis gali būti pateiktas keliose eilutėse) sujungiamas į vieną simbolių eilutę, nepamirštant tarp sujungiamų eilučių įterpti tarpą;

– tekste randami visi dešinieji skliaustai „(“ ir, jei reikia, prieš juos įterpiamas vienas tarpas;

– tekste randami visi kiti skyrybos ženklai ir kairysis skliaustas ir, jei reikia, po jų įterpiama po vieną tarpą; 

– patikrinama, ar kur nors tekste nėra dviejų tarpų greta – radus vienas jų pašalinamas.

procedure tvarkyti (var t: string);

{ sutvarko duotą tekstą }

  var i: integer;

begin

  { pašalinami nereikalingi tarpai }

  i := 1;

  while i < length(t) do

    if (t[i] in ['(', ' ']) and (t[i+1] = ' ')

       then delete (t, i+1, 1)

       else if (t[i] = ' ') and (t[i+1] in ['.', ',',':',';', ')'])

                then delete (t, i, 1)

                else i := i + 1;

  { prieš dešinį skliaustą įterpiamas tarpas }

  i := 2;

  while i <= length(t) do

    if (t[i] = '(') and (t[i-1] <> ' ')

       then insert (' ', t, i)

       else i := i + 1;

  { po ženklų įterpiama po vieną tarpą }

  i := 1;

  while i < length(t) do

    if (t[i] in ['.', ',',':',';', ')']) and (t[i+1] <> ' ')

       then begin

              insert (' ', t, i+1);  { įterpiame tarpą }

              i := i + 1;

            end

       else i := i + 1;

end; { tvarkyti }



Tekstą į fiksuoto dydžio eilutes skaidome taip. Dalindami tekstą į eilutes nekreipiame dėmesio į skyrybos ženklus ir skliaustus: laikome, kad jie yra žodžio, su kuriuo tas ženklas sujungtas (t. y. tarp jų nėra tarpo), dalis. 

Norėdami atskirti naują eilutę, imame (n+1)-ąjį teksto simbolį. Jeigu tai tarpas – eilutę radomę, jei ne – einame tekstu atgal, kol randame tarpą. Teksto dalis iki tarpo bus spausdinama vienoje eilutėje, teksto dalis po tarpo – tolesnėse eilutėse. Tik reikia nepamiršti pašalinti paties tarpo. Priešingu atveju tolesnė eilutė prasidės tarpu.

Taigi, radę teksto dalį, kuri bus spausdinama vienoje eilutėje, suskaičiuojame, kiek papildomai tarpų reikės įterpti, kad tekstas būtų išlygiuotas. 

Pavyzdžiui, į 40 simbolių ilgio eilutę reikia įrašyti tekstą:

Čia(ir(baigiasi(mūsų(metraštis.

Pats tekstas užima 31 simbolį, taigi teks įterpti 9 tarpus. Šiuo metu tekste yra keturi tarpai, taigi naujus tarpus galima įterpinėti į keturias vietas. Apskaičiuojame: 9 div 4 ( 2; 9 mod 4 ( 1. Taigi į vieną vietą reikės įterpti tris tarpus, į tris vietas – po du tarpus:

Čia((((ir(((baigiasi(((mūsų(((metraštis.

procedure skaidyti (n: integer; t: string);

{ suskaido duotą tekstą į eilutes ir atspausdina }

  var viskas: boolean;

      pask, tarpai, i, kiek, liko, nr: integer;

      eil: string;

begin

  viskas := false;

  repeat

    { jei liko paskutinė eilutė, tai ją atspausdiname }

    if length (t) <= n

       then begin

              writeln (t);

              viskas := true

            end

       else begin

              { randame spausdinamos eilutės paskutinį simbolį }

              pask := n + 1;

              while t[pask] <> ' ' do

                pask := pask - 1;

              pask := pask - 1;

              { eil – spausdinimui atskirta eilutė }

              eil := copy (t, 1, pask);

              delete (t, 1, pask+1);

              { randame tarpų skaičių šioje eilutėje }

              tarpai := 0;

              for i := 1 to pask do

                if eil[i] = ' '

                   then tarpai := tarpai + 1;

              { rasime įterpiamų tarpų skaičių }

              kiek := (n - pask) div tarpai;

              liko := (n - pask) mod tarpai;

              nr := 1;

              { sudėliojame tarpus }

              while nr < n do

                begin

                  if eil[nr] = ' '

                     then begin

                            for i := 1 to kiek do

                              insert (' ', eil, nr);

                            nr := nr + kiek;

                            if liko > 0

                               then begin

                                      insert (' ', eil, nr);

                                      liko := liko - 1;

                                      nr := nr + 1

                                    end;

                          end;

                  nr := nr + 1;

                end;

              { atspausdinama  eilutė }

              writeln (eil);

            end;

  until viskas;

  writeln;

end; { skaidyti }

Pirmieji keturi testai tikrina, ar gerai sutvarkomas tekstas (sutvarkytas tekstas telpa į vieną eilutę), penktas ir šeštas testai tikrina, ar pavyko tekstą suskaidyti į duoto ilgio eilutes, paskutinysis – apjungia visus atvejus.






